sum of all terms in submultisets of A000707 Consider the multiset 1,2,2,3,3,3, and A000707.
What is the sum of all terms in the unordered submultisets and in the ordered submultisets as one proceeds through the sequence?  . For k=1 to 4 the sequence begins for unordered submultisets 1,7,42,234 and  the sum of the terms of all ordered submultisets is 1,10,126,1904. https://oeis.org/A000707
 A: Working    with    the    same    notation    as    at    this    MSE
post  we  have  for
the case of ordered multisets the labeled species for $n$ fixed
$$\mathfrak{P_{\le 1}}(\mathcal{U}\mathcal{Z})
\mathfrak{P_{\le 2}}(\mathcal{U}^2\mathcal{Z})
\mathfrak{P_{\le 3}}(\mathcal{U}^3\mathcal{Z})
\cdots
\mathfrak{P_{\le n}}(\mathcal{U}^n\mathcal{Z}).$$
This represents a  row of $m$ available ordered slots  where we select
at most one to contain the value one, at most two to contain the value
two, at most three to contain the  value three and so on and the slots
per value form a set. This yields the exponential generating function
$$G(z) = 
\left.\frac{\partial}{\partial u}
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{u^{qp} z^p}{p!}\right)
\right|_{u=1}
\\ = \left.
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{u^{qp} z^p}{p!}\right)
\sum_{q=1}^n \left(\sum_{p=0}^q \frac{u^{qp} z^p}{p!}\right)^{-1}
\left(\sum_{p=0}^q \frac{qp u^{qp-1} z^p}{p!}\right)
\right|_{u=1}
\\ = 
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{z^p}{p!}\right)
\sum_{q=1}^n \left(\sum_{p=0}^q \frac{z^p}{p!}\right)^{-1}
\left(\sum_{p=0}^q \frac{qp z^p}{p!}\right)
\\ = 
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{z^p}{p!}\right)
\sum_{q=1}^n q\times \left(\sum_{p=0}^q \frac{z^p}{p!}\right)^{-1}
\left(\sum_{p=1}^q \frac{z^p}{(p-1)!}\right)
\\ = 
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{z^p}{p!}\right)
\sum_{q=1}^n q\times \left(\sum_{p=0}^q \frac{z^p}{p!}\right)^{-1}
\left(- \frac{z^{q+1}}{q!} + z \sum_{p=0}^{q} \frac{z^p}{p!}\right)
\\ = 
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{z^p}{p!}\right)
\sum_{q=1}^n q\times 
\left(z - \frac{z^{q+1}}{q!} 
\left(\sum_{p=0}^q \frac{z^p}{p!}\right)^{-1}\right)
\\ = z
\prod_{q=1}^n \left(\sum_{p=0}^q \frac{z^p}{p!}\right)
\sum_{q=1}^n q\times 
\left(1 - \frac{z^{q}}{q!} 
\left(\sum_{p=0}^q \frac{z^p}{p!}\right)^{-1}\right).$$
We are interested in $n! [z^n] G(z)$ which yields the sequence
$$1, 10, 126, 1928, 34630, 713982, 16627912, 431880584, 
\\ 12380457636, 388321923170,\ldots$$
which apparently does not have an OEIS entry yet.
An  implementation goes  like  this (warning,  total enumeration,  not
optimized, tested for values up to twelve)

X :=
proc(n)
option remember;
local gf, res, term, pos;

    if n=1 then return 1 fi;

    gf := expand(add(A[q], q=1..n)^n);

    res := 0;

    for term in gf do
        for pos to n do
            if degree(term, A[pos]) > pos then
                break;
            fi;
        od;

        if pos = n+1 then
            res := res +
            lcoeff(term)*add(q*degree(term, A[q]), q=1..n);
        fi;
    od;

    res;
end;


Z :=
proc(n)
option remember;
local gf;

    gf := z*mul(add(z^p/p!, p=0..q), q=1..n)
    * add(q*(1-z^q/q!/add(z^p/p!, p=0..q)), q=1..n);

    n!*coeff(factor(gf), z, n);
end;

